

Критерієм ламінарності течії є так зване безрозмірне число Рейнольдса, яке дорівнює

$$Re = \frac{\rho v_{max} L}{\eta}. \quad (3.5.31)$$

Наприклад, для води критичне значення числа Рейнольдса приблизно дорівнює $Re_{cr} \approx 2300$. Якщо число Рейнольдса не перевищує критичного значення, течію рідини можна вважати ламінарною і користуватися законом Хагена-Пуазейля.

В'язкість крові помітно вища від в'язкості води, зокрема для кров'яної плазми удвічі, а для нормальної крові приблизно в 4 рази [30], тому хоча її густина лише незначно перевищує густину води, критичне значення числа Рейнольдса для крові помітно нижче, ніж у води. Вважається [30], що прояви турбулентності для кровотоку можливі вже за умови $Re_{cr} \approx 400$ (рис. 3.5.3).




Рис. 3.5.3. Схема виникнення турбулентності потоку крові в судинному шунті

Лабораторна робота 6

Нелінійні моделі математичної фізики

1. Математична модель нелінійного фізичного маятника.
2. Солітони Кортевега-де-Вріза.
 - 2.1. Нелінійне рівняння Кортевега-де-Вріза та його солітонне рішення.
 - 2.2. Мультисолітонні рішення рівняння КдВ та взаємодія пари (дублету) солітонів.

1 Математична модель нелінійного математичного маятника

Фізичний маятник – тверде тіло довільної форми, яке під дією сили тяжіння здійснює коливання навколо нерухомої горизонталь-

ної осі (вісь Y на рис. 3.6.1), яка не проходить через центр маси тіла (точка C на рис. 3.6.1).




Рис. 3.6.1. Кінематика фізичного маятника

Нелінійні властивості фізичного маятника (ФМ) без урахування сил опору можна вивчити за математичною моделлю, яка описується наступним диференціальним рівнянням другого порядку [32]:

$$J\ddot{\varphi} + mgl \cdot \sin(\varphi) = 0, \quad (3.6.1)$$

де J – момент інерції ФМ відносно осі Z обертання маятника; m – маса маятника; l – відстань від центра мас C до осі обертання маятника; g – прискорення вільного падіння, φ – кут відхилення маятника від вертикалі.

Якщо ввести безрозмірний час $\tau = t \sqrt{\frac{mgl}{J}} = \omega_0 t$, де ω_0 – власна частота коливань маятника, то рівняння (3.6.1) можна перевести у безрозмірний вигляд:

```
> restart;
> pm:=diff(phi(tau),tau$2)+sin(phi(tau))=0;
```

$$pm := \frac{d^2}{d\tau^2} \phi(\tau) + \sin(\phi(\tau)) = 0. \quad (3.6.2)$$

У якості початкових умов оберемо дві умови:

- маятник починає рухатися з позиції максимального відхилення від положення рівноваги φ_0 ;
- з нульовою кутовою швидкістю $\dot{\varphi} = 0$:

```
> ic:=phi(0)=phi[0],D(phi)(0)=0;
```

$$ic := \phi(0) = \varphi_0, D(\phi)(0) = 0. \quad (3.6.3)$$

Окрім того, паралельно корисно представити математичну модель (3.6.2) у вигляді системи з двох рівнянь першого порядку:

$$\begin{aligned} & \text{> sys:=diff(phi(tau), tau)=chi(tau),} \\ & \text{diff(chi(tau), tau)+sin(phi(tau))=0;} \\ \text{sys} & := \frac{d}{dt} \phi(\tau) = \chi(\tau), \frac{d}{dt} \chi(\tau) + \sin(\phi(\tau)) = 0. \end{aligned} \quad (3.6.4)$$

Точний формальний розв'язок задачі Коші (3.6.2), (3.6.3) можна отримати засобами Maple:

$$\begin{aligned} & \text{> ans:=dsolve([pm, ic]);} \\ \text{ans} & := \phi(\tau) = \text{RootOf} \left(\begin{array}{c} \phi_0 \\ \frac{1}{\sqrt{2 \cos(_a) - 2 \cos(\phi_0)}} \quad _a + \tau \end{array} \right), \\ \phi(\tau) & = \text{RootOf} \left(\begin{array}{c} _Z \\ \frac{1}{\sqrt{2 \cos(_a) - 2 \cos(\phi_0)}} \quad _a + \tau \\ \phi_0 \end{array} \right). \end{aligned} \quad (3.6.5)$$

Два рішення (3.6.5) відрізняються лише знаками, оскільки межі інтегралів переставлені місцями і не виглядають надто прозоро. Зрештою з (3.6.5) можна зробити висновок, що розв'язку задачі нелінійного фізичного маятника в елементарних функціях не існує. Вираз (3.6.5) дає точний розв'язок у формі інтегрального рівняння.

Проте це не заважає побудувати графіки точного розв'язку засобами команди DEtools[DEplot]. Зобразимо зокрема графіки коливань для двох кутів: $\phi_0 = \frac{2\pi}{3} = 120^\circ$, $\phi_0 = \frac{179\pi}{180} = 179^\circ$.

```
> DEtools[DEplot](pm, phi(tau), tau=-6.25*Pi..6.25*Pi,
[[phi(0)=2*Pi/3, D(phi)(0)=0]], linestyle=dash,
linecolor=blue, numpoints=500):
DEtools[DEplot](pm, phi(tau), tau=-6.25*Pi..6.25*Pi,
[[phi(0)=179*Pi/180, D(phi)(0)=0]], linecolor=red):
> plots[display](%, %, gridlines=true, axes=boxed,
font=[Arial,14], labels=['tau', ' phi ( tau )'],
labelfont=[Arial,14], labeldirections=[horizontal,
vertical], caption=`Рис. 2. Залежність кутів
відхилення від часу`);
```




Рис. 3.6.2. Залежність кутів відхилення від часу

З графіків рис. 3.6.2 можна зробити такі висновки:

- періоди нелінійних коливань, отже, і їх частота, суттєво залежать від їх амплітуди ϕ_0 , зокрема період коливань з амплітудою в 179 град. в декілька разів більший від амплітуди коливань з амплітудою в 120 град;
- коливання з великими амплітудами відхилення істотно відрізняються за формою від гармонічних коливань, характерних для малих амплітуд.

Залежність безрозмірного періоду коливань від амплітуди, тобто відношення періоду нелінійних коливань T до періоду малих

коливань $T_0 = 2\pi\sqrt{\frac{l}{g}}$, дається такою формулою [32]:

```
> Tau:=(4/(2*Pi))*int(1/sqrt(1-
(sin(phi[0]/2))^2*sin(z)^2), z=0..Pi/2);
```

$$T := \frac{2 K\left(\sin\left(\frac{\phi_0}{2}\right)\right)}{\pi}, \quad (3.6.6)$$

де $K(k)$ – спеціальна функція, так званий повний еліптичний інтеграл першого роду. Графік виразу (3.6.6) поданий нижче.

```
> plot([Tau,1], phi[0]=0..Pi, thickness=3,
gridlines=true, axes=boxed, font=[Arial,14],
color=[blue,red], linestyle=[1,4], labels=['phi[0]',
"Період"], labelfont=[Arial,14],
labeldirections=[horizontal, vertical],
caption=`Рис. 3. Залежність безрозмірного періоду
\n від амплітуди нелінійних коливань`);
```




Рис. 3.6.3. Залежність безрозмірного періоду від амплітуди нелінійних коливань

Як видно з рис. 3.6.3, наприклад, період нелінійних коливань для амплітуди $\phi_0 = \frac{7\pi}{8}$ удвічі більший від періоду для малих коливань. Побудуємо також фазові портрети двох нелінійних коливань, показаних на рис. 3.6.2.

```
> DEtools[DEplot]([sys], [phi(tau), chi(tau)],
tau=-6*Pi..6*Pi, [[phi(0)=179*Pi/180,chi(0)=0]],
linecolor=red, numpoints=500):
DEtools[DEplot]([sys], [phi(tau), chi(tau)],
tau=-6*Pi..6*Pi, [[phi(0)=Pi/2, chi(0)=0]],
linecolor=blue, numpoints=500):
> plots[display](%, %, axes=boxed, font=[Arial,14],
labels=['phi', 'chi'], labelfont=[Arial,14],
labeldirections=[horizontal, vertical], caption=
[`Рис. 4. Фазові портрети двох \n нелінійних
коливань`]);
```




Рис. 3.6.4. Фазові портрети двох нелінійних коливань

Фазові портрети показують, що коливання з великою амплітудою суттєво негармонічні, оскільки їх фазова траєкторія помітно відрізняється від еліптичної.

2 Солітони Кортевега-де-Вріза

Існує декілька математичних моделей, для яких солітони є точним розв'язком, із яких найбільш популярними є такі три нелінійних диференціальних рівняння:

- 1) рівняння Кортевега-де Вріза (КдВ);
- 2) нелінійне рівняння Шредінгера (НУШ);
- 3) рівняння синус-Гордона (С-Г).

Розв'язки саме цих рівнянь утворюють три основних типи солітонів:

- солітони Кортевега-де Вріза – КдВ, або так звані акустичні солітони;
- солітони огиначаючої – оптичні солітони;
- топологічні солітони – так звані кінки та анти-кінки.

2.1. Нелінійне рівняння Кортевега-де-Вріза та його солітонне рішення

Однією з найпростіших і найвідоміших моделей, які допускають існування солітонів у своїх рішеннях, є нелінійне диференціальне рівняння Кортевега-де-Вріза. Солітони КдВ, або так звані акустичні солітони, можливі в середовищах з дисперсією, коли швидкість хвилі в середовищі повинна залежати від її довжини, а також суттєвою нелінійністю.

```
> restart:
with(PDEtools): # Необхідні програмні пакети
> KdV:=diff(u(x,t),t) + 6*u(x,t)*diff(u(x,t),x) +
diff(u(x,t),x$3);
```

$$KdV := \frac{\partial}{\partial t} u(x, t) + 6 u(x, t) \left(\frac{\partial}{\partial x} u(x, t) \right) + \frac{\partial^3}{\partial x^3} u(x, t) . \quad (3.6.7)$$

Солітон є рухомою хвилею, тому доцільно шукати солітонні рішення рівняння (3.6.7) у вигляді рухомої хвилі із застосуванням таких функцій як гіперболічні секанс та косеканс, які описують його форму [33], [34]:

```
> TWSolutions(KdV, singsol=false,
functions=[sech,csch]);
```

$$\left\{ u(x, t) = 2 _C2^2 \operatorname{sech}(_C2 x + _C3 t + _C1)^2 - \frac{4 _C2^3 + _C3}{6 _C2} \right\},$$

$$\left\{ u(x, t) = -2 _C2^2 \operatorname{csch}(_C2 x + _C3 t + _C1)^2 - \frac{4 _C2^3 + _C3}{6 _C2} \right\} . \quad (3.6.8)$$

Перевіримо, що (3.6.8) дійсно є рішенням рівняння (3.6.7) за допомогою підстановки функції (3.6.8) у рівняння (3.6.7), користуючись командою перевірки пакету PDEtools:

$$\begin{aligned} & | > \text{pdetest}(\%[1], \text{KdV}); \\ & | \text{pdetest}(\%[2], \text{KdV}); \\ & \qquad \qquad \qquad 0, \qquad \qquad \qquad (3.6.9) \\ & \qquad \qquad \qquad 0. \end{aligned}$$

Два отриманих рішення відповідають солітонам один із яких розповсюджується в напрямку осі Ox , а інший – у протилежному напрямі. Надалі розглядатимемо лише перший із них, оскільки їх властивості абсолютно тотожні, за виключенням знаку швидкості, який визначається напрямом руху:

$$\begin{aligned} & | > u(x, t) := - \\ & | (1/6) * (_C3 + 4*_C2^3) / _C2 + 2*_C2^2 * \text{sech}(_C1 + _C2*x + _C3*t) ^ \\ & | 2; \\ u(x, t) & := 2_C2^2 \text{sech}(_C2 x + _C3 t + _C1)^2 - \frac{4_C2^3 + _C3}{6_C2}. \end{aligned} \quad (3.6.10)$$

Через те, що солітон є рухомою хвилею він мусить задовольняти хвильовому рівнянню, отже його фазова швидкість може бути обчислена через відношення других похідних від функції (3.6.10) за координатами та часом:

$$\begin{aligned} & | > \text{simplify}(\text{diff}(u(x, t), x^2)): \\ & | \text{simplify}(\text{diff}(u(x, t), t^2)): \\ & | v = -\text{sqrt}(\%/%%, \text{symbolic}); \\ & \qquad \qquad \qquad v = -\frac{C3}{C2}, \qquad \qquad \qquad (3.6.11) \end{aligned}$$

$$\begin{aligned} & | > u(x, t) := \text{factor}(\text{eval}(u(x, t), _C3 = -v*_C2)); \\ u(x, t) & := 2_C2^2 \text{sech}(-v_C2 t + _C2 x + _C1)^2 - \frac{2_C2^2}{3} + \frac{v}{6}. \end{aligned} \quad (3.6.12)$$

Незалежну від координат та часу частину рішення (3.6.12) звичайно покладають нульовою, хоча б тому, що вона визначає лише рівень відліку для функції $u(x, t)$. Отже:

$$\begin{aligned} & | > _C2 = \text{sqrt}(v/4); \\ & \qquad \qquad \qquad _C2 = \frac{\sqrt{v}}{2}, \qquad \qquad \qquad (3.6.13) \end{aligned}$$

$$\begin{aligned} & | > u(x, t) := \text{eval}(u(x, t), _C2 = \text{sqrt}(v/4)); \\ u(x, t) & := \frac{v \text{sech}\left(-\frac{v^{3/2} t}{2} + \frac{\sqrt{v} x}{2} + _C1\right)^2}{2}. \end{aligned} \quad (3.6.14)$$

Що до константи $_C1$, то її можна включити в координату або час у якості точки відліку, припустимо таким способом:

$$\begin{aligned} & | > _C1 = -\text{sqrt}(v) * \chi / 2; \\ & \qquad \qquad \qquad _C1 = -\frac{\sqrt{v} \chi}{2}, \qquad \qquad \qquad (3.6.15) \end{aligned}$$

```
> u(x,t) := eval(u(x,t), _C1=-sqrt(v)*chi/2);
```

$$u(x,t) := \frac{v \operatorname{sech}\left(\frac{v^{3/2}t}{2} - \frac{\sqrt{v}x}{2} + \frac{\sqrt{v}\chi}{2}\right)^2}{2}. \quad (3.6.16)$$

Або в дещо простішому вигляді з двома параметрами v , χ – швидкість та початок відліку координат:

```
> u:=unapply(simplify(v*(sech(sqrt(v)*(x-chi-v*t)/2))^2/2), x, t, v, chi);
```

$$u := (x, t, v, \chi) \mapsto \frac{v}{2 \cosh\left(\frac{\sqrt{v}(vt + \chi - x)}{2}\right)^2}. \quad (3.6.17)$$

Зобразимо функцію (3.6.17) з фіксованими модельними параметрами: $v=0.5$, $\chi=0$.

```
> plot3d(u(x,t,0.5,0), x=-12..12, t=-8..8,
axes=framed, projection=orthogonal,
orientation=[98,40,10], style=patch,
lightmodel=light3, grid=[48,32], font=[Arial,14],
labels=["x", "t", "u(x, t, 0.5, 0)"],
labelfont=[Arial,14], labeldirections=[horizontal,
horizontal, vertical], caption="Рис. 5. Солітон КдВ:
часово-просторова форма");
```




Рис. 3.6.5. Солітон КдВ: часово-просторова форма

А також у вигляді анімації, яка представляє рух солітону у часі на поверхні каналу, параметри ті ж самі, але їх можна змінювати, досліджуючи різні солітони.

```
> plots[animate](plot, [0.1+u(x,t,0.5,0), x=-16..16,
filled=true, gridlines=true, axes=boxed, color=blue],
t=-36..36, frames=36, title='', font=[Helvetica,14],
labels=["x", "u(x, t, 0.5, 0)"], label-
font=[Arial,14], labeldirections=[horizontal,
vertical]); # caption="Рис. 6. Солітон у русі на
поверхні каналу"
```





Рис. 3.6.6. Солітон у русі на поверхні каналу. Показано перший кадр у початковий момент часу

2.2. Мультисолітонні рішення рівняння КдВ та взаємодія пари (дублету) солітонів

Характерною особливістю рівняння КдВ є те, що воно припускає не лише прості солітонні рішення у вигляді окремого солітону, яке було розглянуте вище, але також мультисолітонні рішення у вигляді сукупності (мультиплету) пов'язаних між собою солітонів.

Розглянемо для прикладу солітонний дублет, тобто пару солітонів [33], [34]. Такий дублет у дублеті пов'язані між собою нелінійним способом. Солітонний дублет є точним аналітичним рішенням рівняння КдВ і має такий вигляд:

```
> restart:
sol2:=unapply(2*(v2-v1)*(v1*sech(sqrt(v1/2)*
(x-2*v1*t))^2+v2*csch(sqrt(v2/2)*
(x-2*v2*t))^2)/(sqrt(2*v1)*tanh(sqrt(v1/2)*
(x-2*v1*t))-sqrt(2*v2)*coth(sqrt(v2/2)*
(x-2*v2*t))),v1,v2,x,t);
```

$$sol2 := (v1, v2, x, t) \mapsto \left(2 (v2 - v1) \left(v1 \operatorname{sech} \left(\frac{\sqrt{2} \sqrt{v1} (-2 v1 t + x)}{2} \right)^2 + v2 \operatorname{csch} \left(\frac{\sqrt{2} \sqrt{v2} (-2 v2 t + x)}{2} \right)^2 \right) \right) / \left(\left(\sqrt{2} \sqrt{v1} \tanh \left(\frac{\sqrt{2} \sqrt{v1} (-2 v1 t + x)}{2} \right) - \sqrt{2} \sqrt{v2} \coth \left(\frac{\sqrt{2} \sqrt{v2} (-2 v2 t + x)}{2} \right) \right)^2 \right), \quad (3.6.18)$$

де v_1, v_2 – швидкості кожного солітону з дублету, які пропорційні їх амплітудам, отже, чим швидшим є солітон тим він вищий і вузчий. Покажемо просторову-часову еволюцію пари солітонів у дублеті графічно:

```
> plot3d(sol2(1/3,2/3,x,t), x=-20..20, t=-20..20,
axes=framed, grid=[200,200], orientation=[45,45,0],
font=[HELVETICA,14], lightmodel=`light4`, color=gold,
labels=["x", "t", "Амплітуда"], labelfont=[Arial,14],
labeldirections=[horizontal, horizontal, vertical],
title=`Рис. 7. Дублет солітонів: \n еволюція в часі та
просторі`);
```





Рис. 3.6.7. Дублет солітонів: еволюція в часі та просторі

Інший погляд, так би мовити «згори» на еволюцію солітонного дублету дозволяють графіки густини, на яких світлішими виглядають вищі ділянки фігури на рис. 3.6.8, причому солітон із вищою амплітудою виглядає світлішим:

```
> plots[densityplot](sol2(1/3,2/3,x,t), x=-20..20,
t=-20..20, axes=framed, style=patchnogrid,
font=[HELVETICA,14], labels=["x", "t"],
labelfont=[Arial,14], labeldirections=[horizontal,
vertical], title="Рис. 8. Дублет солітонів:\n еволюція
та зіткнення в точці (0,0)");
```



**Рис. 3.6.8. Дублет солітонів:
еволюція та зіткнення в точці $(0, 0)$**

Рис. 3.6.8 показує, що до моменту часу $t=0$ x -координати солітону з меншою амплітудою, який зображено темнішою та ширшою смугою, випереджають координати солітону з більшою амплітудою, він на рис. 3.6.8 світліший та вузчий. Отже, менший солітон випереджав більшого. Після зіткнення солітонів у точці $(x,t)=(0,0)$ солітони міняються місцями: тепер уже більший солітон випереджає меншого.

Зверніть увагу, лінії, які зображують солітони, після зіткнення не є прямим продовженням їх же ліній до зіткнення, спостерігається візуально помітне зміщення. Втім, найбільш наочними завжди є анімації процесів. Розглянемо анімацію процесу зіткнення двох солітонів, представленою на рис. 3.6.8, та описаного вище:

```
> plots[animate](sol2(1/3,2/3,x,t), x=-50..50,
t=-30..30, frames=100, color=blue, thickness=3,
gridlines=true, numpoints=280, axes=boxed,
font=[HELVETICA,14], labels=["x", "Амплітуда"],
labelfont=[Arial,14], labeldirections=[horizontal,
vertical], title=`Рис. 9. Взаємодія двох солітонів у
дублеті`);
```




Рис. 3.6.9. Взаємодія двох солітонів у дуплеті.
(показано перший кадр у початковий момент часу)

Анімація рис. 3.6.9 ясно показує взаємодію двох солітонів: швидший солітон доганяє повільніший. Під час зіткнення вони «обмінюються амплітудами», передній солітон стає вищим та вужчим, а задній навпаки – нижчим та ширшим. Зверніть увагу, форма солітонів до і після зіткнення залишається однаковою. Лише безпосередньо в момент зіткнення обидва солітони формують об'єднаний комплекс. Ситуація дуже нагадує зіткнення двох пружних куль із різними імпульсами на більярдному столі.

Власна назва солітонів завдячує саме цій обставині – ці відокремлені хвилі під час зіткнень поведуться як частинки.

Лабораторна робота 7

Приклади економіко-математичних моделей

1. Приклади типових економіко-математичних задач.
 - 1.1. Задача про оптимальний розподіл капіталу.
 - 1.2. Коефіцієнт Джині та його обчислення.
2. Виробнича функція з прикладами застосування.
 - 2.1. Виробнича функція.
 - 2.2. Задача оптимального розподілу часу на роботу та дозвілля.